PROGRAM NMP.ACSL

!PBPK MODEL FOR N-METHYL PYRROLIDONE

!FINAL RAT MODEL (5/09)

!T.S. POET,P HINDERLITER. CHEMICAL DOSIMETRY GROUP, PNNL, RICHLAND, WA

!MODEL TRANSFERRED FROM SIMUSOLV TO ACSLXTREME FORMAT IN 08

!MODEL CONFIGURED FOR INHALATION (OPEN, WHOLE BODY/NOSE ONLY)

! IV, ORAL, DERMAL, AND IP ROUTES OF ADMINISTRATION. 

!MODEL TRACKS DISPOSITION OF NMP AND 5-HNMP. 

!ASSUMPTIONS: 

! (1) FLOW-LIMITED (ALL COMPARTMENTS) 

! (2) METABOLISM OF NMP BY A SAT PATHWAY TO FORM 5HNP

! (3) METABOLISM OF HNP BY SATURABLE PATHWAY TO ETC. 

! (5) METABOLISM OCCURS ONLY IN THE LIVER 

! (6) TISSUE:BLOOD PART. COEFF. = HUMAN = KRISHNAN EQN 

! UPDATED IN CMD FILE TO MEASURED IN-HOUSE

! (7) 5HNP ELIMIN FROM MIXED VENOUS - 1ST ORDER

! THIS DIFFERS FROM 02: URINE BY *GFR CLEARANCE FROM KIDNEY

! METAB RATE CONST. FROM REPORT - UPDATED WITH LIT VALUES IN CMD FILE

! PREG ADDED - OTHER PARAMETERS CHANGED NOMINALLY TO HARMONIZE WITH 

FETAL IPA MODEL OF 

! GENTRY ET AL. REGU TOX PHARM 36:51-68, 2002

!

! Further updates by Paul Schlosser, US EPA in August and September 2013 

and December 2014

INITIAL

! MODEL UNITS

! CONCENTRATION, MG/L

! FLOW, L/HR

! BODY WT, KG

CONSTANT BWINIT=0. ! PRE-PREGNANCY BODY WEIGHT (KG)

CONSTANT RATS=1. !NUMBER OF ANIMALS IN EXPT

CONSTANT MWNMP=99.13 !MOL. WT. NMP, MG/MMOL

CONSTANT MWHP= 116.14 !MOL. WT. 5-HNP, MG/MMOL

!BLOOD FLOWS

!FROM BROWN ET AL TOX IND HEALTH 97 

!AND/OR FROM IPA MODEL OF GENTRY ET AL., 

! BLOOD FLOWS (FRACTION OF CARDIAC OUTPUT)

 CONSTANT QCC = 0 ! CARDIAC OUTPUT (L/HR FOR 1 KG ANIMAL)

 CONSTANT QPC = 0 ! ALVEOLAR VENT. RATE

 

 CONSTANT QFATC = 0 ! FAT (NON-PREGNANT)

 CONSTANT QLIVC = 0 ! LIVER

 CONSTANT QMAMC = 0 ! MAMMARY TISSUE (NON-PREGNANT)

 CONSTANT QSKNC = 0 ! SKIN

 CONSTANT QUTRC = 0 ! UTERUS (NON-PREGNANT)

 CONSTANT QRAPC = 0 ! RAPID USE STATIC RAPID FOR RATS (MUST BE 

CHANGED FOR HUMAN)

 
! PERMEABILITY-AREA PRODUCT (L/HR)

 CONSTANT PAFC = 0.1 ! DIFFUSION ON FETAL SIDE OF PLACENTA

! TISSUE VOLUMES (FRACTION OF BODY WEIGHT)

!FROM BROWN ET AL TOX IND HEALTH 97 FOR RATS

!OR FROM GENTRY ET AL

 CONSTANT VLUC = 0 ! LUNG

 CONSTANT VFATC = 0 ! FAT (NON-PREGNANT)

 CONSTANT VLIVC = 0 ! LIVER

 CONSTANT VMAMC = 0 ! MAMMARY TISSUE (NON-PREGNANT)

 CONSTANT VRAPC = 0 ! RAPIDLY PERFUSED

 CONSTANT VUTRC = 0 ! UTERUS (NON-PREGNANT)

 CONSTANT VBLC = 0 ! TOTAL BLOOD

 

! FOR PARENT MODEL, SKIN COMPARTMENT IS ONLY DEFINED AS DOSED SKIN

 CONSTANT VSKC = 0.19 ! SKIN 

 CONSTANT SA = 0.01 !SURFACE AREA EXPOSED, SQ.CM

 TSA = 906.*BWINIT**(2./3.) !TOTAL BODY SURFACE AREA, SQ.CM.

 !MCDOUGAL ET AL. T.A.P. 

85(1996)286

 IF (CONCL.GT.0.0) THEN

VSKCC = VSKC*SA/TSA

 QSKCC = QSKNC*SA/TSA

 ELSE

VSKCC = VSKC*SA/TSA

 QSKCC = QSKNC*SA/TSA

 ENDIF

! SLOWLY PERFUSED (DEFINED AS BALANCE OF TISSUES AND FLOWS)

VSC = 0.91 - (VLUC + VFATC + VLIVC + VMAMC + VRAPC + VUTRC + VBLC + 

VSKCC)

! NOTE: 0.91 IS APPROX WHOLE BODY LESS BONE

QSC = 1. - (QFATC + QLIVC + QMAMC + QRAPC + QUTRC + QSKCC)

! SCALED BLOOD FLOWS (L/HR)

 QCINIT = QCC * (BWINIT**0.75)

 QFATI = QFATC * QCINIT

 QLIV = QLIVC * QCINIT

 QMAMI = QMAMC * QCINIT

 QRAP = QRAPC * QCINIT

 QSKN = QSKCC * QCINIT

 QSLW = QSC * QCINIT

 QUTRI = QUTRC * QCINIT

! SCALED TISSUE VOLUMES (L)

 VLU = VLUC * BWINIT

 VFATI = VFATC * BWINIT

 VLIVI = VLIVC * BWINIT

 VRAP = VRAPC * BWINIT

 VSLW = VSC * BWINIT

 VMAMI = VMAMC * BWINIT

 VUTRI = VUTRC * BWINIT

 VSK = VSKCC * BWINIT
 VBL = VBLC * BWINIT ! TOTAL BLOOD

 VA = 0.25*VBL !ARTERIAL BLOOD

 VV = 0.75*VBL !VENOUS BLOOD

! PREGNANCY PARAMETERS

 CONSTANT NUMFET = 7.0 ! NUMBER OF FETUSES 

 CONSTANT PUPBW = 4500. ! BIRTH WEIGHT (MG)

 CONSTANT VFETD18 = 1051.254 ! VOLUME OF FETUS AT DAY 18 ( ) OF 

PREGNANCY

! CONVERSION FACTORS

 CONSTANT MGKG = 1.0E6 ! CONVERSION FACTOR FROM MG TO KG

!PARTITION COEFFICIENTS

!EXPERIMENTALLY MEASURED VALUES

 CONSTANT PB=0. !NMP BLOOD:AIR

 CONSTANT PF=0 !NMP FAT:BLOOD - MEASURED

 CONSTANT PL=0 !MEASURED

 CONSTANT PR=0 !MEASURED LIVER

 CONSTANT PS=0 !NOT MEASURED MUSCLE - CORRECTED FOR FILTER 

ERROR USING SKIN PROPORTIONALITY

 CONSTANT PSKL=0 !MEASURED

 CONSTANT PLU=0 !NMP LUNG:BLOOD

 CONSTANT PSKA= 0 !NMP SKIN:AIR 

CONSTANT PSKB=0 !NMP SKIN:BLOOD

 CONSTANT PM=0 !MAMMARY, ESTIMATED FORM LIVER

 CONSTANT PPLA=0

 CONSTANT PUTR=0

 

!EXPERIMENTALLY MEASURED VALUES

 CONSTANT PLHNP=0 !LIVER MEASURED

 CONSTANT PBHNP=0 !ESTIMATED AVG OF "OTHER" TISSUES

 CONSTANT PFHNP=0 !MEASURED

 CONSTANT PPLHNP=0

 

!METABOLIC RATE CONSTANTS

 CONSTANT KM=0 !MICHAELIS CONSTANT, MG/L

 CONSTANT VMAXC=0 !MAX. ENZ. ACT., MG/HR/L

 VMAX1 = VMAXC*BWINIT**0.75

 

 !5HNP TO OTHER METABS

 CONSTANT KM2=0 !MICHAELIS CONSTANT, MG/L

 CONSTANT VMAX2C=0 !MAX. ENZ. ACT., MG/HR/L

 VMAX2 = VMAX2C*BWINIT**0.75

 

!URINARY ELIMINATION OF 5-HNMP - CLEARED FROM BLOOD

 CONSTANT KLC=0

 KL=KLC/(BWINIT**0.25)

CONSTANT KLNC=0 !URINARY LOSS OF NMP, L/HR

KLN=KLNC/(BWINIT**0.25)

!FRACTIONAL ABSORPTION 
 CONSTANT FRACIN = 1 !FRACTIONAL UPTAKE OF NMP BY INHAL,START AT 

65%

 !OF ALVEOLAR - AS IN AKESSON ET AL 

1997

 CONSTANT FRACOR = 1.0 !FRACTION ABSORBED ORALLY, INITALLY 100%

 CONSTANT FRACF=1

! INITIAL CONDITIONS FOR CLOSED CHAMBER INHALATION

 CONSTANT VCHC = 9E9 ! VOLUME OF CLOSED CHAMBER (L),START LARGE FOR 

OPEN

 CONSTANT KLOSS = 0.0 ! CHAMBER LOSS RATE /HR

 

 

!TIMING COMMANDS

 

CONSTANT TCHNG=6.0 !END OF INHAL EXPOSURE, HR

 CONSTANT TSTOP=24.0 !END OF EXPERIMENT/SIMULATION, HR

 CONSTANT MAXT=0.01 !MAXIMUM STEP SIZE, HR

 CONSTANT MINT=1E-7

 CONSTANT CINT = 0.2 !DATA LOGGING RATE /HR

 CONSTANT GDDAYS=0.0 ! OFFSET FOR GESTATIONAL DAY SIMULATION

 CONSTANT GDMONTHS=0.0 !OFFSET FOR HUMAN GD SIMULATION

!INITIAL EXPOSURE CONDITIONS

 ! EXPOSURE CONDITIONS BASED ON USER DEFINED INITIAL AMOUNTS OF CHEMICAL 

(MG)

 CONSTANT CONCPPM = 0.0 !AIR CONCENTRATION IN PPM!

constant concmgs = 0.0

! Used to set air conc'n as mg/m3

 VCH = VCHC-(RATS*BWINIT) !VOLUME OF OCCUPIED CHAMBER

 CONCMG = CONCMGS/1000 + CONCPPM*MWNMP/24451. !CONVERT PPM TO 

MG/LITER

 CONSTANT DOSEINTERVAL=24 !TIME 

BETWEEN DAILY DOSES

constant concchppm0 = 0

! Initial ppm in closed chamber

conchmg0= concchppm0*MWNMP/24451.

 ACHO = conchmg0 * VCH !INIT. AMT IN CHAMBER, MG! 

 !ORAL

 CONSTANT KAS=1.0 !1ST ORDER RATE CONST FOR ORAL ABS from 

stomach,HR-1

CONSTANT KAI=1 !1ST 

ORDER RATE CONST FOR ORAL ABS from intestines,HR-1

 CONSTANT KSI=1

 CONSTANT DOSE=0.0 !ORAL DOSE IN MG/KG BW

 ODOSE = FRACOR*DOSE*BWINIT !CONVERT MG/KG BW TO MG 

TOTAL(ORAL)

! ODOSE multiplied by FRACOR to reduce oral 

bioavailability

constant dose2=0.0 ! ORAL 

Dose in mg/kg BW, but total dose increases w/ BW

gavds=dose2*FRACOR*BWINIT ! Initial value for this 

dose
 !FEED

 CONSTANT KASF=1.0 !1ST ORDER RATE CONST FOR ORAL ABS,HR-1

 CONSTANT DOSEF=0.0 !ORAL DOSE IN MG/KG BW in feed

 !IV 

 CONSTANT IVDOSE=0.0 !IV DOSE, MG/KG NMP

 !DERMAL

 CONSTANT CONCL = 0.0 !CONC OF NMP IN LIQUID, MG/L

 CONSTANT KPL = 0.0 !PERM COEFF FOR LIQUID, CM/HR

 CONSTANT VLIQ = 1.0E-99 !INITIAL VOLUME APPLIED, L

 CONSTANT DENSITY= 1.03

constant DSK=0.0 ! Initial amount (mg/kg BW) rubbed into 

skin

ASKO=DSK*BWINIT

constant GDSTOP=15 ! Last GD of dermal dosing

constant twash=8.0 ! Wash time in Becci et al. 

(1982) exposures

CONSTANT FAD=0.78 !FRAC no absorbed in Payan et al

 !IN VITRO HUMAN VAN DYK ET AL. AIHA J 56: 651-660

 !START WITH SMALL SA SO VSKE IS NON-ZERO (USED IN DENOMINATOR OF 

CSK CALCULATION)

 

 !IP

 CONSTANT IPDOSE = 0.0 !IP DOSE, MG/KG NMP

 CONSTANT KIP=1.0 !1ST ORDER RATE OF ABS, HR-1

 PDOSE = IPDOSE*BWINIT !TOTAL IP DOSE, MG

!DOSING SCHEDULE

if (DSK.GT.0.0) then

schedule SKWASH.AT.TWASH

ENDIF

 SCHEDULE OFFD.AT.TCHNG !TURN OFF EXPOSURE AT TCHNG

 CIZONE = 1.0 !START WITH INHALATION ON

 IVZONE = 1.0 !START WITH IV ON

IF (CONCL.GT.0.0) THEN

 DZONE = 1.0 !START WITH DERMAL ON

ELSE

DZONE = 0.0

ENDIF

constant tstart=0.2 ! offset start-time for gavage dosing

schedule GAVD.at.TSTART

 ALGORITHM IALG=2 !GEAR ALGORITHM 

 

END

DYNAMIC

 

DERIVATIVE

!===============FETAL AND BW CHANGES W/PREGNANCY=======================

 GDMONTH=GDMONTHS/0.64 !TO WAG GROWTH

 HOURS = T

 MINUTES = T * 60.0

 DAYS = T / 24.0 + GDDAYS +GDMONTH

! VOLUME OF FAT (L)
 VFAT = VFATI * (1.0 + (0.0165 * DAYS))

! VOLUME OF FETUS (KG)

 IF (DAYS.LT.10.0) THEN

 VFET = (1.0e-8 + NUMFET * ((0.1206 * DAYS)**4.53)) / MGKG

 ELSE IF (DAYS.LT.17.0) THEN

 VFET = (1.0e-8 + NUMFET * ((1.5 * (DAYS - 9))**2.8)) / MGKG

 ELSE

 VFET = (1.0e-8 + NUMFET * (VFETD18 + (((PUPBW - VFETD18) / 4.0) 

* (DAYS - 17)))) / MGKG

 ENDIF

! VOLUME OF MAMMARY TISSUE (L)

 VMAM = VMAMI * (1.0 + (0.27 * DAYS))

! VOLUME OF PLACENTA (L)

 IF (DAYS.LT.6.0) THEN

 VPLA = 1.0e-8

 ELSE IF (DAYS.LT.10.0) THEN

 VPLA = (1.0e-8 + NUMFET * (8. * (DAYS - 6.))) / MGKG

 ELSE 

 VPLA = (1.0e-8 + NUMFET * ((32 * EXP(-0.23 * (DAYS - 10)))+ (40 

* (EXP(0.28 * (DAYS - 10)) - 1)))) / MGKG

 ENDIF

! VOLUME OF UTERUS (L)

 IF (DAYS.LE.3.0) THEN

 VUTR = VUTRI

 ELSE

 VUTR = VUTRI * (1.0 + (0.077 * ((DAYS - 3.)**1.6)))

 ENDIF

!VOLUME OF LIVER INCREASE !Corley et al CRC 03,BUELKE-SAM ET AL '82 AND 

OTHERS

IF (DAYS.LT.5.0) THEN

VLIV=VLIVI

ELSE 

VLIV= VLIVI * (1.0 + (0.0455 * ((DAYS - 5.0))))

ENDIF

! INCREASE IN BODY WEIGHT (KG)

 BW = BWINIT + (VFAT - VFATI) + VFET + (VMAM - VMAMI) + 

VPLA + (VUTR - VUTRI)+(VLIV - VLIVI)

 

! SCALED ALVEOLAR VENTILATION (L/HR)

 QP = QPC * ((BW-VFET-VPLA)**0.75)

 

! INCREASE IN BLOOD FLOWS (L/HR)

 QFAT = QFATI * (VFAT / VFATI)

 QMAM = QMAMI * (VMAM / VMAMI)

 QUTR = QUTRI * (VUTR / VUTRI)

! TOTAL BODY FOR HNMP

 QB = QRAP+QSLW+QSKN+QMAM+QUTR ! 
 VB = VRAP+VSLW+VLU+VSK+VMAM+VUTR !

! BLOOD FLOW TO PLACENTA (L/HR)

 IF (DAYS.LT.6.0) THEN

 QPLA = 0.0

 ELSE IF (DAYS.LT.10.0) THEN

 QPLA = (NUMFET * (0.55 * (DAYS - 6.0))) / 24

 ELSE IF (DAYS.LE.12.0) THEN

 QPLA = (NUMFET * (2.2 * EXP(-0.23 * (DAYS - 10)))) / 24

 ELSE 

 QPLA = (NUMFET * ((2.2 * EXP(-0.23 * (DAYS - 10)))+ ((0.1207 * 

(DAYS - 12.0))**4.36))) / 24

 ENDIF

! INCREASED CARDIAC OUTPUT (L/HR)

 QC = QFAT+QLIV+QSLW+QRAP+QSKN+QMAM+QPLA+QUTR

! SCALED PERMEABILITY-AREA PRODUCT

 PAF = PAFC * (VFET**0.75)

!==================FIRST MODEL FOR TRACKING NMP=========================

!EQUATIONS FOR ORAL GAVAGE DOSING 

!note this is an oral model that expects some direct absorption to liver 

and some transfered to intestine for absorption to liver

!this structure explains early peak and slow elimination observed in 

Midgely oral exposure data. Validated by Ghantous data

 RAOA = -(KAS * AO)-(KSI*AO) 

 AO = ODOSE+ INTEG(RAOA,0.0) !AMT REMAINING TO BE ABS, MG

 RAO=KAS*AO

 OABS = INTEG(RAO,0.0) 

RINTC=(KSI*AO)-(KAI*AINTC) !RATE OF CHANGE IN INTESTINES

AINTC=INTEG(RINTC,0.0)

RAINTEST=KAI*AINTC !TRANSFER TO LIVER

OIBS=INTEG(RAINTEST,0.0)

!EQUATIONS FOR FEED DOSING 

 FDOSE = DOSEF*FRACF*BW !CONVERT MG/KG BW TO MG TOTAL(ORAL)

 RAF = KASF * AF !*FRACF

 AF = FDOSE - INTEG(RAF,0.0) !AMT REMAINING TO BE ABS, MG!

 FABS = INTEG(RAF,0.0) 

!AL = AMOUNT NMP IN LIVER COMPARTMENT (MG) 

 RAL = QLIV*(CA - CVL)+ RAIP + RAO + RAF - RAML+RAINTEST

 AL = INTEG(RAL, 0.0)

 CVL = AL/(VLIV*PL)

 

 RAML = (VMAX1*CVL)/(KM+CVL) !SATURABLE METABOLISM, MG/HR

 AML = INTEG(RAML,0.0) !AMT NMP METAB BY SATURABLE PATH, MG

 AML1B = RATS*AML*MWHP/MWNMP !TOT AMT HNP PRODUCED IN LIVER, MG

!EQUATIONS FOR IP DOSING 

 RAIP = KIP * AIP

 AIP = INTEG(-RAIP,PDOSE) !AMT REMAINING TO BE ABS, MG!
 IPABS = INTEG(RAIP,0.0) 

 

!EQUATIONS FOR IV INFUSION 

 IVR = IVZONE*IVDOSE*BW/Tchng !RATE OF INFUSION, MG/HR using Tchng

 TIV = INTEG(IVR,0.0) !TOTAL AMOUNT INJECTED, MG

! ARTERIAL BLOOD 

 RAAB = (QC * (CVLU - CA))-RAUNP

 AAB = INTEG(RAAB, 0.0) !AMOUNT, MG

 CA = AAB / VA !CONCENTRATION, MG/L

 AAUCB = INTEG(CA, 0.0) !AUC, HR*MG/L

RAUNP = KLN*CA*VV !FIRST ORDER RATE OF LOSS (URINE

AUNP = INTEG(RAUNP,0.0)

! CHAMBER CONCENTRATION (MG/L)

 RACH = (RATS * QP * CLEX) - (FRACIN * RATS * QP * CI) - (KLOSS * 

ACH)

 ACH = INTEG(RACH, ACHO)

 

! THE FOLLOWING CALCULATION YIELDS AN AIR CONCENTRATION EQUAL TO 

! THE CLOSED CHAMBER VALUE IF A CLOSED CHAMBER RUN IS IN PLACE AND

! A SPECIFIED CONSTANT AIR CONCENTRATION IF AN OPEN CHAMBER RUN IS IN 

PLACE

 CCH = (ACH / VCH)! * CIZONE) + (CONCMG * (1.0 - CLON))

 

 CCPPM = CCH *24451/MWNMP

 CLOSS = INTEG(KLOSS * ACH,0.0)

CI = CCH*PULSE(0., DOSEINTERVAL,TCHNG) + CIZONE*CONCMG ! MG/L

! LUNGS

 RALU = (QP * ((FRACIN * CI) - CLEX)) + RVV - (QC * CVLU)

 ALU = INTEG(RALU, 0.0)

 CLU = ALU / VLU !CONCENTRATION, MG/L

 CVLU = CLU / PLU !EXITING CONCENTRATION, MG/L

! AMOUNT INHALED 

 RINH = FRACIN * QP * CCH *CIZONE

 AINH = INTEG(RINH, 0.0) ! MG PER 

 AINHC = AINH * RATS ! MG FOR A GROUP OF RATS

! AMOUNT EXHALED 

 CLEX = CV / PB ! CONCENTRATION, MG/L

 RAEX = QP * CLEX

 AEX = INTEG(RAEX, 0.0) ! AMOUNT, MG PER 

 AEXC = AEX * RATS ! AMOUNT, MG, FOR A GROUP OF RATS

!ASK = AMOUNT NMP IN SKIN TISSUES (MG) AND DERMAL DOSING 

 RASK = QSKN*(CA - CSKV) + RADL

 ASK = INTEG(RASK,ASKO) ! Initial value, ASKO, added for Becci 

et al. (1982) exposures

 CSK = ASK/VSK !'NMP IN SKIN, MG/L'
CSKV = CSK/PSKB ! NMP IN VENOUS 

BLOOD

 CVSK3 = CSK*1000/MWNMP !'NMP IN CVSK, MICROMOL/L'

 

CONCL2=CONCL*FAD

CSURF=(CONCL2-(ADL/VLIQ))*DZONE

RADL=(KPL*SA/1000.0)*((CSURF-(CSK/PSKL))*DZONE - (1.0-DZONE)*(CSK/PSKA)) 

! 

! 2ND term, (1.0-DZONE)*(CSK/PSKA), allows for evaporative loss when 

DZONE=0

ADL=INTEG(RADL,0.0)

! NOTE - NO LOSS TERM. TRY WITHOUT OR ADD LOSS UP-FRONT BY 

SUBTRACTING

! AMOUNT RECOVERED FOR EACH STUDY WITH AMOUNT (CONC) ORIGINALLY 

APPLIED

! "LOSS" OR STICKING PROBABLY ESSENTIALLY IMMEDIATE AND NOT 

KINETIC

! REPORTS OF ~11-24% STICKING TO DRESSING

! AMOUNT IN FAT (MG)

 RAFAT = QFAT * (CA - CVFAT)

 AFAT = INTEG(RAFAT, 0.0)

 CFAT = AFAT / VFAT

 CVFAT = CFAT / PF

! AMOUNT IN FETUSES (MG)

 RAFET = PAF * (CPLA - CFET)

 AFET = INTEG(RAFET, 0.0)

CFET = AFET / VFET

 AUCCFET = INTEG(CFET, 0.0)

! AMOUNT IN UTERUS (MG)

 RAUTR = QUTR * (CA - CVUTR)

 AUTR = INTEG(RAUTR, 0.0)

 CUTR = AUTR / VUTR

 CVUTR = CUTR / PUTR

! AMOUNT IN MAMMARY TISSUE (MG)

 RAMAM = QMAM * (CA - CVMAM)

 AMAM = INTEG(RAMAM, 0.0)

 CMAM = AMAM / VMAM

 CVMAM = CMAM / PM

 

! AMOUNT IN PLACENTA (MG)

 RAPLA = (QPLA * (CA - CVPLA)) + (PAF * (CFET - CPLA))

 APLA = INTEG(RAPLA, 0.0)

CPLA = APLA / VPLA 

 CVPLA = CPLA / PPLA

!AS = AMOUNT IN SLOWLY PERFUSED TISSUES (MG) 

 RAS = QSLW*(CA - CVS)

 AS = INTEG(RAS, 0.0)
 CVS = AS/(VSLW*PS)

 CS = AS/VSLW

!AR = AMOUNT IN RAPIDLY PERFUSED TISSUES (MG) 

 RAR = QRAP*(CA - CVR)

 AR = INTEG(RAR, 0.0)

 CVR = AR/(VRAP*PR)

 CR = AR/VRAP

!MIXED VENOUS BLOOD

RVV = QC*CV

RV=(QFAT*CVFAT+QLIV*CVL+QSLW*CVS+QRAP*CVR+QSKN*CSKV+CVMAM*QMAM+CVPL

A*QPLA+QUTR*CVUTR+IVR)-RVV

AV=INTEG(RV,0.0)

CV=AV/VV

AUCBB=INTEG(CV,0.0) !AUC, HR*MG/L

 

!-----------MASS BALANCE NMP --------------

 BODY = (AFAT+AR+AS+AL+ASK+AV+ALU+AAB+APLA+AMAM+AUTR)

 TMASS = RATS*(BODY + AML + AEX+AUNP+AFET)!COMPARE TO 

 !AINH FOR OC MASS BAL

!OR OABS FOR ORAL MASS BAL

!OR TIV FOR IV MASS BAL

!OR ADL FOR DERMAL LIQUID

MASBAL=TMASS/(AINH+OABS+TIV+ADL+OIBS+1E-9)

! CHECK BLOOD FLOWS

 QTOT = QFATI + QLIV + QRAP + QSKN + QSLW + QUTRI +QMAM+QPLA

 QRECOV = 100.0 * (QTOT / QC)

!===============SECOND MODEL FOR TRACKING HNP=====================

!ALHP = AMOUNT HNMP IN LIVER COMPARTMENT (MG) 

 RALHP = QLIV*(CAHP-CVLHP)+ RAML1 - RAMLH

 RAML1=RAML*MWHP/MWNMP

 AML2B=INTEG(RAML1,0.0)

 ALHP = INTEG(RALHP,0.0) !AMT IN MG HNMP, CORRECTED FOR MW

 CVLHP = ALHP/(VLIV*PLHNP) !TOTAL HNMP

 

 RAMLH = (VMAX2*CVLHP)/(KM2+CVLHP) !SATURABLE METABOLISM, MG/HR

 AMLH = INTEG(RAMLH,0.0) !AMT HNMP METAB BY SATURABLE PATH, MG

rdose=ramlh/(BW**0.75)

tdose=integ(rdose,0.0)

!ABHP = AMOUNT HNMP IN TISSUES (MG) 

 RABHP = QB*(CAHP - CBSHP)

 ABHP = INTEG(RABHP,0.0)

 CBSHP = ABHP/(VB*PBHNP)

 

!AFHP = AMOUNT HNMP IN FAT (MG) 

 RFSHP = QFAT*(CAHP - CVFHP)

 AFHP = INTEG(RFSHP,0.0)

 CVFHP = AFHP/(VFAT*PFHNP)
!CVHP = MIXED VENOUS BLOOD CONC TOTAL HNMP (MG/L) 

 CRHP = (QLIV*CVLHP + QB*CBSHP + QFAT*CVFHP + QPLA*CVPLHP)-QC*CVHP-RAUHP

 AVHP = INTEG (CRHP,0.0)

 CVHP = AVHP/VBL

 CAHP = CVHP

 CVHP2 = CVHP*1000/MWHP !VENOUS BLOOD TOT CONC HNMP IN MICROM

 AUCVHP = INTEG(CVHP2,0.0) !AUC HNMP VEN. BLOOD, MICROMOL*HR/L

 

 ! AMOUNT IN PLACENTA (MG)

 RAPLHP = (QPLA * (CAHP - CVPLHP)) + (PAF * (CFETHP - CPLHP))

 APLHP = INTEG(RAPLHP, 0.0)

CPLHP = APLHP / VPLA

 CVPLHP = CPLHP / PPLHNP

 

 ! AMOUNT IN FETUSES (MG)

 RAFETHP = PAF * (CPLHP - CFETHP)

 AFETHP = INTEG(RAFETHP, 0.0)

CFETHP = AFETHP / VFET

 AUCFETHP = INTEG(CFETHP, 0.0)

 

!RATE OF ELIM IN THE URINE, RAUHP, FROM MIXED BLOOD

 RAUHP = KL*CAHP*VA !FIRST ORDER RATE

 AUHP = INTEG(RAUHP,0.0) !CUMULATIVE AMT HNMP IN URINE (MG), NOT 

MGEQ

 

!-----------MASS BALANCE--------------

!-----------MASS BALANCE 5-HNMP SUBMODEL--------------

 BODYHP = (AFHP+ABHP+ALHP+AVHP+AFETHP+APLHP)*RATS

 TMASHP = RATS*(AUHP + BODYHP +AMLH) !COMPARE TO AML1B

! CHECK BLOOD FLOWS 5HNMP COMPARTMENT

 QTOTH = QLIV + QFAT + QB+QPLA

 QRECOVH = 100.0 * (QTOTH / QC)

TERMT(T .GE. TSTOP) !----STATEMENT TO STOP EXECUTION---

END !END OF DERIVATIVE

! The following discrete block allows for repeated gavage dosing, but 

with 

! the total dose (gavds) only updated every 3 days, per the protocol of 

! Becci et al. (1982) and Saillenfait et al. (2002); PMS 9-16-13

discrete GAVD

IF (ROUND(DAYS).EQ.9.0) gavds=FRACOR*dose2*BW

IF (ROUND(DAYS).EQ.12.0) gavds=FRACOR*dose2*BW

IF (ROUND(DAYS).EQ.15.0) gavds=FRACOR*dose2*BW

IF (ROUND(DAYS).EQ.18.0) gavds=FRACOR*dose2*BW

ODOSE=ODOSE+gavds

if (DAYS.LT.GDSTOP) schedule GAVD .at. (T+DOSEINTERVAL)

end

!EXPOSURE CONTROL

DISCRETE SKWASH
ASK = 0.0 ! Assume skin washing in Becci et al. (1982) removes all 

NMP from skin

if (DAYS.LT.GDSTOP) SCHEDULE REAPPLY.AT.(T+DOSEINTERVAL-TWASH)

END

DISCRETE REAPPLY

IF (ROUND(DAYS).EQ.9.0) ASKO=DSK*BW

IF (ROUND(DAYS).EQ.12.0) ASKO=DSK*BW

IF (ROUND(DAYS).EQ.15.0) ASKO=DSK*BW

IF (ROUND(DAYS).EQ.18.0) ASKO=DSK*BW

ASK = ASK + ASKO

SCHEDULE SKWASH.AT.(T+TWASH)

END

DISCRETE OFFD

 IVZONE=0.0 !TURN IV OFF

 CIZONE=0.0 !TURN INHAL EXPOSURE OFF

 DZONE=0.0 !TURN OFF DERMAL

SCHEDULE OND.AT.(T+DOSEINTERVAL-TCHNG)

END

DISCRETE OND

 CIZONE=1.0 !TURN INHAL EXPOSURE ON

SCHEDULE OFFD.AT.(T+TCHNG)

END

END !END OF DYNAMIC

END !END OF PROGRAM